D01BBF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

D01BBF returns the weights and abscissae appropriate to a Gaussian quadrature formula with a specified number of abscissae. The formulae provided are Gauss–Legendre, Gauss–Rational, Gauss–Laguerre and Gauss–Hermite.

2 Specification

```
SUBROUTINE D01BBF(D01XXX, A, B, ITYPE, N, WEIGHT, ABSCIS, IFAIL)
INTEGER             ITYPE, N, IFAIL
real                A, B, WEIGHT(N), ABSCIS(N)
EXTERNAL            D01XXX
```

3 Description

This routine returns the weights and abscissae for use in the Gaussian quadrature of a function $f(x)$. The quadrature takes the form

$$S = \sum_{i=1}^{n} w_i f(x_i)$$

where w_i are the weights and x_i are the abscissae (see Davis and Rabinowitz [1], Froberg [2], Ralston [3] or Stroud and Secrest [4]).

Weights and abscissae are available for Gauss–Legendre, Gauss–Rational, Gauss–Laguerre and Gauss–Hermite quadrature, and for a selection of values of n (see Section 5).

(a) Gauss–Legendre Quadrature:

$$S \simeq \int_a^b f(x)\,dx$$

where a and b are finite and it will be exact for any function of the form

$$f(x) = \sum_{i=0}^{2n-1} c_i x^i$$

(b) Gauss–Rational quadrature:

$$S \simeq \int_a^\infty f(x)\,dx \quad (a+b>0) \quad \text{or} \quad S \simeq \int_{-\infty}^a f(x)\,dx \quad (a+b<0)$$

and will be exact for any function of the form

$$f(x) = \sum_{i=2}^{2n+1} \frac{c_i}{(x+b)^i} = \frac{\sum_{i=0}^{2n-1} c_{2n+1-i}(x+b)^i}{(x+b)^{2n+1}}$$

(c) Gauss–Laguerre quadrature, adjusted weights option:

$$S \simeq \int_a^\infty f(x)\,dx \quad (b>0) \quad \text{or} \quad S \simeq \int_{-\infty}^a f(x)\,dx \quad (b<0)$$

and will be exact for any function of the form

$$f(x) = e^{-bx} \sum_{i=0}^{2n-1} c_i x^i$$

(d) Gauss–Hermite quadrature, adjusted weights option:

$$S \simeq \int_{-\infty}^{+\infty} f(x)\,dx$$

and will be exact for any function of the form

$$f(x) = e^{-b(x-a)^2} \sum_{i=0}^{2n-1} c_i x^i \quad (b > 0)$$

(e) Gauss–Laguerre quadrature, normal weights option:

$$S \simeq \int_a^{\infty} e^{-bx} f(x)\,dx \quad (b > 0) \quad \text{or} \quad S \simeq \int_{-\infty}^{a} e^{-bx} f(x)\,dx \quad (b < 0)$$

and will be exact for any function of the form

$$f(x) = \sum_{i=0}^{2n-1} c_i x^i$$

(f) Gauss–Hermite quadrature, normal weights option:

$$S \simeq \int_{-\infty}^{+\infty} e^{-b(x-a)^2} f(x)\,dx$$

and will be exact for any function of the form:

$$f(x) = \sum_{i=0}^{2n-1} c_i x^i$$

Note. That the Gauss–Legendre abscissae, with $a = -1$, $b = +1$, are the zeros of the Legendre polynomials; the Gauss–Laguerre abscissae, with $a = 0$, $b = 1$, are the zeros of the Laguerre polynomials; and the Gauss–Hermite abscissae, with $a = 0$, $b = 1$, are the zeros of the Hermite polynomials.

4 References

[1] Davis P J and Rabinowitz P (1975) *Methods of Numerical Integration* Academic Press

[2] Froberg C E (1970) *Introduction to Numerical Analysis* Addison–Wesley

[3] Ralston A (1965) *A First Course in Numerical Analysis* McGraw–Hill 87–90

[4] Stroud A H and Secrest D (1966) *Gaussian Quadrature Formulas* Prentice–Hall

5 Parameters

1: D01XXX — SUBROUTINE, supplied by the NAG Fortran Library. *External Procedure*

The name of the routine indicates the quadrature formula:

D01BAZ, for Gauss–Legendre weights and abscissae;
D01BAY, for Gauss–Rational weights and abscissae;
D01BAX, for Gauss–Laguerre weights and abscissae;
D01BAW, for Gauss–Hermite weights and abscissae.

The name used must be declared as EXTERNAL in the (sub)program from which D01BBF is called.

In certain implementations, to avoid name clashes between single and double precision versions, names of auxiliary routines have been changed, e.g., D01BAX to BAXD01. Please refer to the Users' Note for your implementation.

2: A — ***real*** *Input*

3: B — ***real*** *Input*

On entry: the quantities a and b as described in the appropriate subsection of Section 3.

4: ITYPE — INTEGER *Input*

On entry: indicates the type of weights for Gauss–Laguerre or Gauss–Hermite quadrature (see Section 3):

if ITYPE = 1, adjusted weights will be returned;
if ITYPE = 0, normal weights will be returned.

Constraint: ITYPE = 0 or 1.

For Gauss–Legendre or Gauss–Rational quadrature, this parameter is not used.

5: N — INTEGER *Input*

On entry: the number of weights and abscissae to be returned, n.

Constraint: N = 1,2,3,4,5,6,8,10,12,14,16,20,24,32,48 or 64.

6: WEIGHT(N) — ***real*** array *Output*

On exit: the N weights. For Gauss–Laguerre and Gauss–Hermite quadrature, these will be the adjusted weights if ITYPE = 1, and the normal weights if ITYPE = 0.

7: ABSCIS(N) — ***real*** array *Output*

On exit: the N abscissae.

8: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

The N-point rule is not among those stored. If the soft fail option is used, the weights and abscissae returned will be those for the largest valid value of N less than the requested value, and the excess elements of WEIGHT and ABSCIS (i.e., up to the requested N) will be filled with zeros.

IFAIL = 2

The value of A and/or B is invalid.

Gauss–Rational: $A + B = 0$
Gauss–Laguerre: $B = 0$
Gauss–Hermite: $B \leq 0$

If the soft fail option is used the weights and abscissae are returned as zero.

IFAIL = 3

Laguerre and Hermite normal weights only: underflow is occurring in evaluating one or more of the normal weights. If the soft fail option is used, the underflowing weights are returned as zero. A smaller value of N must be used; or adjusted weights should be used (ITYPE = 1). In the latter case, take care that underflow does not occur when evaluating the integrand appropriate for adjusted weights.

7 Accuracy

The weights and abscissae are stored for standard values of A and B to full machine accuracy.

8 Further Comments

Timing is negligible.

9 Example

This example program returns the abscissae and (adjusted) weights for the six-point Gauss–Laguerre formula.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     D01BBF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          N
      PARAMETER        (N=6)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      real             A, B
      INTEGER          IFAIL, ITYPE, J
*     .. Local Arrays ..
      real             ABSCIS(N), WEIGHT(N)
*     .. External Subroutines ..
      EXTERNAL         D01BAX, D01BBF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'D01BBF Example Program Results'
      A = 0.0e0
      B = 1.0e0
      ITYPE = 1
      IFAIL = 0
*
      CALL D01BBF(D01BAX,A,B,ITYPE,N,WEIGHT,ABSCIS,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'Laguerre formula,', N, ' points'
      WRITE (NOUT,*)
      WRITE (NOUT,*) '    Abscissae        Weights'
      WRITE (NOUT,*)
      WRITE (NOUT,99999) (ABSCIS(J),WEIGHT(J),J=1,N)
      STOP
*
99999 FORMAT (1X,2e15.6)
99998 FORMAT (1X,A,I3,A)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
D01BBF Example Program Results

Laguerre formula,  6 points

    Abscissae        Weights

   0.222847E+00   0.573536E+00
   0.118893E+01   0.136925E+01
   0.299274E+01   0.226068E+01
   0.577514E+01   0.335052E+01
   0.983747E+01   0.488683E+01
   0.159829E+02   0.784902E+01
```